This worksheet deals with the problem of parametrizing systems of partial differential equations. We treat a system of partial differential equations with non-constant coefficients that appears in the study of the Lie algebra SU(2). See C. M. Bender, G. V. Dunne, L. R. Mead, *Underdetermined systems of partial differential equations*, Journal of Mathematical Physics, vol. 41 no. 9 (2000), pp. 6388-6398.

```
> with(Ore_algebra):
> with(OreModules):
```

We define *Alg* as the Weyl algebra, i.e., D_i acts as differential operator w.r.t. x_i, i=1, 2, 3.

```
> Alg := DefineOreAlgebra(diff=[D[1],x[1]], diff=[D[2],x[2]],
> diff=[D[3],x[3]], polynom=[x[1],x[2],x[3]]):
```

We enter the system matrix R:

```
> R := evalm([[x[3]*D[1]-x[1]*D[3], x[3]*D[2]-x[2]*D[3], -1],
> [-1, x[1]*D[2]-x[2]*D[1], x[1]*D[3]-x[3]*D[1]],
> [x[2]*D[1]-x[1]*D[2], -1, x[2]*D[3]-x[3]*D[2]]]);
```

$$R := \begin{bmatrix} x_3\,\mathrm{D}_1 - x_1\,\mathrm{D}_3 & x_3\,\mathrm{D}_2 - x_2\,\mathrm{D}_3 & -1 \\ -1 & x_1\,\mathrm{D}_2 - x_2\,\mathrm{D}_1 & x_1\,\mathrm{D}_3 - x_3\,\mathrm{D}_1 \\ x_2\,\mathrm{D}_1 - x_1\,\mathrm{D}_2 & -1 & x_2\,\mathrm{D}_3 - x_3\,\mathrm{D}_2 \end{bmatrix}$$

Next, we define the formal adjoint *R_adj* of R:

```
> R_adj := Involution(R, Alg);
```

$$R\_adj := \begin{bmatrix} x_1\,\mathrm{D}_3 - x_3\,\mathrm{D}_1 & -1 & x_1\,\mathrm{D}_2 - x_2\,\mathrm{D}_1 \\ x_2\,\mathrm{D}_3 - x_3\,\mathrm{D}_2 & x_2\,\mathrm{D}_1 - x_1\,\mathrm{D}_2 & -1 \\ -1 & x_3\,\mathrm{D}_1 - x_1\,\mathrm{D}_3 & x_3\,\mathrm{D}_2 - x_2\,\mathrm{D}_3 \end{bmatrix}$$

Applying *Exti* to *R_adj*, we check whether the system given by the matrix R is parametrizable or not:

```
> st := time(): Ext1 := Exti(R_adj, Alg, 1): time() - st; Ext1[1];
> Ext1[2];
```

$$0.409$$

$$\begin{bmatrix} x_2\,\mathrm{D}_3 - x_3\,\mathrm{D}_2 & 0 & 0 \\ x_1\,\mathrm{D}_3 - x_3\,\mathrm{D}_1 & 0 & 0 \\ x_1\,\mathrm{D}_2 - x_2\,\mathrm{D}_1 & 0 & 0 \\ 0 & x_2\,\mathrm{D}_3 - x_3\,\mathrm{D}_2 & 0 \\ 0 & x_1\,\mathrm{D}_3 - x_3\,\mathrm{D}_1 & 0 \\ 0 & x_1\,\mathrm{D}_2 - x_2\,\mathrm{D}_1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$\begin{bmatrix} x_1 & x_2 & x_3 \\ \mathrm{D}_1 & \mathrm{D}_2 & \mathrm{D}_3 \\ -1 & x_1\,\mathrm{D}_2 - x_2\,\mathrm{D}_1 & x_1\,\mathrm{D}_3 - x_3\,\mathrm{D}_1 \end{bmatrix}$$

We see that the torsion submodule $\mathrm{t}(M)$ of the left *Alg*-module M which is associated with the system is non-trivial. More precisely, we find in *Ext1*[1] the operators that kill the corresponding torsion elements in *Ext1*[2]: Applying any of the operators in the ith column of *Ext1*[1] to the ith row of *Ext1*[2] gives a zero row modulo the equations of the system given by R, i.e., gives zero in the module M. We conclude that the system of partial differential equations given by R is *not* parametrizable.

By using *TorsionElements*, we can obtain the same generating set of the torsion submodule but written in terms of the dependent variables *F1*, *G1*, *H1* of the system. The first matrix gives the relations that the torsion elements θ_i satisfy, i=1, 2, the second matrix defines θ_i in terms of *F1*, *G1*, *H1*:

```
> TorsionElements(R,
> [F1(x[1],x[2],x[3]),G1(x[1],x[2],x[3]),H1(x[1],x[2],x[3])], Alg);
```

$$\left[\left[\begin{array}{l} -x_3\,(\frac{\partial}{\partial x_2}\,\%2) + x_2\,(\frac{\partial}{\partial x_3}\,\%2) = 0 \\ -x_3\,(\frac{\partial}{\partial x_1}\,\%2) + x_1\,(\frac{\partial}{\partial x_3}\,\%2) = 0 \\ -x_2\,(\frac{\partial}{\partial x_1}\,\%2) + x_1\,(\frac{\partial}{\partial x_2}\,\%2) = 0 \\ -x_3\,(\frac{\partial}{\partial x_2}\,\%1) + x_2\,(\frac{\partial}{\partial x_3}\,\%1) = 0 \\ -x_3\,(\frac{\partial}{\partial x_1}\,\%1) + x_1\,(\frac{\partial}{\partial x_3}\,\%1) = 0 \\ -x_2\,(\frac{\partial}{\partial x_1}\,\%1) + x_1\,(\frac{\partial}{\partial x_2}\,\%1) = 0 \end{array}\right], \left[\begin{array}{c} \%2 = x_1\,\mathrm{F1}(x_1,\,x_2,\,x_3) + x_2\,\mathrm{G1}(x_1,\,x_2,\,x_3) + x_3\,\mathrm{H1}(x_1,\,x_2,\,x_3) \\ \%1 = (\frac{\partial}{\partial x_1}\,\mathrm{F1}(x_1,\,x_2,\,x_3)) + (\frac{\partial}{\partial x_2}\,\mathrm{G1}(x_1,\,x_2,\,x_3)) + (\frac{\partial}{\partial x_3}\,\mathrm{H1}(x_1,\,x_2,\,x_3)) \end{array}\right]\right]$$

$$\%1 := \theta_2(x_1,\,x_2,\,x_3)$$
$$\%2 := \theta_1(x_1,\,x_2,\,x_3)$$

Ext1[3] provides a parametrization of the torsion-free part $M/t(M)$ of M:

```
> Ext1[3];
```

$$\left[\begin{array}{c} x_3\,\mathrm{D}_2 - x_2\,\mathrm{D}_3 \\ x_1\,\mathrm{D}_3 - x_3\,\mathrm{D}_1 \\ x_2\,\mathrm{D}_1 - x_1\,\mathrm{D}_2 \end{array}\right]$$

Let us point out that we recover the same parametrization as in (Bender, Dunne, Mead, 2000) (up to a mistake made in (Bender, Dunne, Mead, 2000, p. 6394) concerning the existence of the torsion elements):

```
> ApplyMatrix(Ext1[3], [H(x[1],x[2],x[3])], Alg);
```

$$\left[\begin{array}{c} x_3\,(\frac{\partial}{\partial x_2}\,\%1) - x_2\,(\frac{\partial}{\partial x_3}\,\%1) \\ -x_3\,(\frac{\partial}{\partial x_1}\,\%1) + x_1\,(\frac{\partial}{\partial x_3}\,\%1) \\ x_2\,(\frac{\partial}{\partial x_1}\,\%1) - x_1\,(\frac{\partial}{\partial x_2}\,\%1) \end{array}\right]$$
$$\%1 := \mathrm{H}(x_1,\,x_2,\,x_3)$$

Let us continue with the structural analysis of the torsion-free part $M/t(M)$ of M. Of course, $M/t(M)$ has trivial torsion submodule:

```
> Exti(Involution(Ext1[2], Alg), Alg, 1);
```

$$\left[\left[\begin{array}{ccc} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{array}\right], \left[\begin{array}{ccc} x_1 & x_2 & x_3 \\ \mathrm{D}_1 & \mathrm{D}_2 & \mathrm{D}_3 \\ -1 & x_1\,\mathrm{D}_2 - x_2\,\mathrm{D}_1 & x_1\,\mathrm{D}_3 - x_3\,\mathrm{D}_1 \end{array}\right], \left[\begin{array}{c} x_3\,\mathrm{D}_2 - x_2\,\mathrm{D}_3 \\ x_1\,\mathrm{D}_3 - x_3\,\mathrm{D}_1 \\ x_2\,\mathrm{D}_1 - x_1\,\mathrm{D}_2 \end{array}\right]\right]$$

We check for reflexivity of the torsion-free part $M/t(M)$:

```
> Exti(Involution(Ext1[2], Alg), Alg, 2);
```

$$\left[\left[\begin{array}{c} x_2\,\mathrm{D}_3 - x_3\,\mathrm{D}_2 \\ x_1\,\mathrm{D}_3 - x_3\,\mathrm{D}_1 \\ x_1\,\mathrm{D}_2 - x_2\,\mathrm{D}_1 \end{array}\right], \left[\begin{array}{c} 1 \end{array}\right], \mathrm{SURJ}(1)\right]$$

We see that the torsion-free part is not a reflexive left *Alg*-module.

Let us finish by checking whether the system given by R is parametrizable or not if we turn the variables x_1, x_2, x_3 into invertible elements of *Alg*. More precisely, we consider the system to be given over the Ore algebra B_3, which differs from the Weyl algebra A_3 only in the fact that the domain of coefficients of D_1, D_2, D_3 is not the polynomial ring in x_1, x_2, x_3, but the field of rational functions in x_1, x_2, x_3. However, in *OreModules*, we use $Alg = A_3$ as above but, instead of *Exti*, we apply *ExtiRat* which takes into account that the coefficients are rational functions:

```
> ExtiRat(R_adj, Alg, 1);
```

$$\left[\left[\begin{array}{cc} x_3\,\mathrm{D}_2 - x_2\,\mathrm{D}_3 & 0 \\ x_3\,\mathrm{D}_1 - x_1\,\mathrm{D}_3 & 0 \\ 0 & x_3\,\mathrm{D}_2 - x_2\,\mathrm{D}_3 \\ 0 & x_3\,\mathrm{D}_1 - x_1\,\mathrm{D}_3 \end{array}\right], \left[\begin{array}{ccc} x_1 & x_2 & x_3 \\ 0 & -x_2 - {x_1}^2\,\mathrm{D}_2 + x_2\,x_1\,\mathrm{D}_1 & -x_3 - {x_1}^2\,\mathrm{D}_3 + x_1\,x_3\,\mathrm{D}_1 \end{array}\right],\right.$$

$$\left.\left[\begin{array}{c} x_3\,\mathrm{D}_2 - x_2\,\mathrm{D}_3 \\ x_1\,\mathrm{D}_3 - x_3\,\mathrm{D}_1 \\ x_2\,\mathrm{D}_1 - x_1\,\mathrm{D}_2 \end{array}\right]\right]$$

Again, we conclude that the system is not parametrizable because non-trivial torsion elements exist. A generating set of torsion elements is given in *Ext1*[2] and it can also be obtained in this case by using *TorsionElementsRat* (i.e., the version of *TorsionElements* that copes with rational coefficients):

```
> TorsionElementsRat(R_adj,
> [F1(x[1],x[2],x[3]),G1(x[1],x[2],x[3]),H1(x[1],x[2],x[3])], Alg);
```

$$\left[\left[\begin{array}{c} x_3\,(\frac{\partial}{\partial x_2}\,\%4) - x_2\,(\frac{\partial}{\partial x_3}\,\%4) = 0 \\ x_3\,(\frac{\partial}{\partial x_1}\,\%4) - x_1\,(\frac{\partial}{\partial x_3}\,\%4) = 0 \\ x_3\,(\frac{\partial}{\partial x_2}\,\%1) - x_2\,(\frac{\partial}{\partial x_3}\,\%1) = 0 \\ x_3\,(\frac{\partial}{\partial x_1}\,\%1) - x_1\,(\frac{\partial}{\partial x_3}\,\%1) = 0 \end{array}\right],\right.$$

$$\left.\left[\begin{array}{c} \%4 = x_3\,\mathrm{F1}(x_1,\,x_2,\,x_3) + x_1\,\%3 + x_2\,\%2 \\ \%1 = x_1\,\%3 + {x_3}^2\,(\frac{\partial}{\partial x_1}\,\%3) - x_3\,x_1\,(\frac{\partial}{\partial x_3}\,\%3) + x_2\,\%2 + {x_3}^2\,(\frac{\partial}{\partial x_2}\,\%2) - x_2\,x_3\,(\frac{\partial}{\partial x_3}\,\%2) \end{array}\right]\right]$$

$\%1 := \theta_2(x_1,\,x_2,\,x_3)$
$\%2 := \mathrm{H1}(x_1,\,x_2,\,x_3)$
$\%3 := \mathrm{G1}(x_1,\,x_2,\,x_3)$
$\%4 := \theta_1(x_1,\,x_2,\,x_3)$